

# DRAFT: Transnumber Derivations and Model

Norbert Völker, University of Essex

October 27, 2006

## Contents

<b>1</b>	<b>Transarithmetic: Axiomatic Classes</b>	<b>1</b>
1.1	New constants: xor, infinity, sgn . . . . .	2
1.2	Axiomatic class trans_add . . . . .	2
1.3	Axiomatic class trans_mult (Axioms A12-A28) . . . . .	3
1.4	Axiomatic classes trans_complete and trans_reals . . . . .	4
<b>2</b>	<b>Transarithmetic Theorems</b>	<b>4</b>
2.1	Very elementary equations . . . . .	5
2.2	Distinctness of six basic constants: 0, nullity, +1,-1, plus/minus infinity . . . . .	6
2.3	sign equivalences . . . . .	9
2.4	Algebraic stuff and inequalities... . . . .	9
2.5	Closure of reals under addition . . . . .	13
2.6	Closure of reals under uminus . . . . .	14
2.7	Closure of positive reals under inverse . . . . .	19
2.8	Closure of reals under multiplication . . . . .	20
<b>3</b>	<b>A Model for Transarithmetic</b>	<b>21</b>
<b>4</b>	<b>A model for axiomatic class trans_add</b>	<b>22</b>
4.1	Distinctness of special values . . . . .	23
<b>5</b>	<b>Transnumber ordering</b>	<b>26</b>
5.1	Lattice-completeness of trans_numbers . . . . .	29
<b>6</b>	<b>A model for axiomatic class trans_mult</b>	<b>30</b>
6.1	Inverse and division . . . . .	34

## 1 Transarithmetic: Axiomatic Classes

**theory** *TransNumberAclass*

**imports** *Main Real*

**begin**

### 1.1 New constants: xor, infinity, sgn

Note: This "list" xor is different from binary xor when there are three or more arguments.

**consts**

*xor* :: *bool list*  $\Rightarrow$  *bool*

**primrec**

*xor* [] = *False*

*xor* (*a* # *x*) = ((*xor* *x*  $\wedge$   $\neg$  *a*) | ( $\neg$  (*list-ex id* *x*)  $\wedge$  *a*))

**lemma** *xor-singleton*: *xor* [*b*] = *b*

**by** *simp*

**lemma** *xor-pair*: *xor* [*a*,*b*] = (*a*  $\neq$  *b*)

**by** *auto*

**lemma** *xor-triple*:

*xor* [*a*,*b*,*c*] = ((*a*  $\wedge$   $\neg$  *b*  $\wedge$   $\neg$  *c*)  $\vee$  ( $\neg$  *a*  $\wedge$  *b*  $\wedge$   $\neg$  *c*)  $\vee$  ( $\neg$  *a*  $\wedge$   $\neg$  *b*  $\wedge$  *c*))

**by** *auto*

**axclass** *infinity* < *type*

Note: there is no separate constant minus-infinity

**consts**

*infinity* :: '*a*::*infinity* ( $\infty$  100)

**axclass** *nullity* < *type*

**consts**

*nullity*:: '*a* ::*nullity* ( $\Phi$ )

**axclass** *sgn* < *zero,one,minus, ord*

**consts**

*sgn* :: '*a* :: *sgn*  $\Rightarrow$  '*a*

**axclass** *trans-sgn* < *sgn, nullity*

*trans-sgn*:

*sgn* *a* = (if 0 < *a* then 1 else

if 0 = *a* then 0 else

if *a* < 0 then - 1 else

(\* *a* =  $\Phi$  \*)  $\Phi$ )

### 1.2 Axiomatic class trans\_add

**axclass**

*trans-add < zero, infinity, nullity, plus, minus*

$$A1: a + (b + c) = (a + b) + c$$

$$A2: a + b = b + a$$

$$A3: 0 + a = a$$

$$A4: \Phi + a = \Phi$$

$$A5: \llbracket a \neq -\infty; a \neq \Phi \rrbracket \implies \infty + a = \infty$$

$$A6: a - b = a + (-b)$$

$$A7: -(-a) = a$$

$$A8: \llbracket a \neq \infty; a \neq -\infty; a \neq \Phi \rrbracket \implies a - a = 0$$

$$A9: -\Phi = \Phi$$

$$A10: \llbracket a \neq \infty; a \neq \Phi \rrbracket \implies a - \infty = -\infty$$

$$A11: \infty - \infty = \Phi$$

**instance** *trans-add*  $\subseteq$  *comm-monoid-add*

**apply** (*intro-classes*)

**apply** (*rule A1 [THEN sym]*)

**by** (*rule A2, rule A3*)

### 1.3 Axiomatic class *trans\_mult* (Axioms A12-A28)

**axclass**

*trans-mult < trans-add, trans-sgn, one, times, inverse, ord*

$$A12: a * (b * c) = (a * b) * c$$

$$A13: a * b = b * a$$

$$A14: 1 * a = a$$

$$A15: \Phi * a = \Phi$$

$$A16: \infty * 0 = \Phi$$

$$A17: a / b = a * \text{inverse } b$$

$$A18: \llbracket a \neq 0; a \neq \infty; a \neq -\infty; a \neq \Phi \rrbracket \implies a / a = 1$$

$$A19: a \neq -\infty \implies \text{inverse } (\text{inverse } a) = a$$

$$A20: \text{inverse } 0 = \infty$$

$$A21: \text{inverse } (-\infty) = 0$$

$$A22: \text{inverse } \Phi = \Phi$$

$$A23: (\infty * a = \infty) = (0 < a)$$

$$A24: (\infty * a = -\infty) = (a < 0)$$

$$A25: 0 < \infty$$

$$A26: (0 < a - b) = (b < a)$$

$$A27: (a > b) = (b < a)$$

$$A28: \text{xor } [a < 0, a = 0, 0 < a, a = \Phi]$$

$$A29: \neg ((a = \infty \vee a = -\infty) \wedge \text{sgn } b \neq \text{sgn } c \wedge (b + c \notin \{0, \Phi\})) \\ \implies a * (b+c) = (a * b) + (a * c)$$

$$A30: a \leq b = (a = b \vee a < b)$$

**instance** *trans-mult*  $\subseteq$  *comm-monoid-mult*  
**by** (*intro-classes*, *rule A12* [*THEN sym*], *rule A13*, *rule A14*)

## 1.4 Axiomatic classes *trans\_complete* and *trans\_reals*

**constdefs**

*lattice-complete* :: ('a::ord) set  $\Rightarrow$  bool  
*lattice-complete* xs ==  
 $\forall$  ys. ys  $\subseteq$  xs  $\longrightarrow$  ( $\exists$  u  $\in$  xs. ( $\forall$  y  $\in$  ys. y  $\leq$  u)  
 $\wedge$  ( $\forall$  v  $\in$  xs. ( $\forall$  y  $\in$  ys. y  $\leq$  v)  $\longrightarrow$  u  $\leq$  v))

**axclass** *trans-complete* < *trans-add*, *one*, *times*, *inverse*, *ord*  
*A31*: *lattice-complete* {x. x  $\neq$   $\Phi$ }

**axclass** *trans-reals* < *trans-mult*, *trans-complete*

TODO: validate definition by proving lattice-completeness of [0..1]

**end**

## 2 Transarithmetic Theorems

**theory** *TransNumberDerivations*

**imports** *TransNumberAclass*

**begin**

**declare** *A3[simp]* *A4[simp]* *A7[simp]* *A9[simp]* *A11[simp]*  
*A14[simp]* *A15[simp]* *A16[simp]*  
*A20[simp]* *A21[simp]* *A22[simp]* *A25[simp]*

**constdefs**

*reals* :: ('a::trans-reals) set

$reals == \{x. x \neq \Phi \wedge x \neq \infty \wedge x \neq -\infty\}$

Note : following subclassing allows reuse of standard `add_ac`, etc.

**instance** *trans-reals*  $\subseteq$  *comm-monoid-add* ..

## 2.1 Very elementary equations

**lemma** *additive-identity-right*[*simp*]:  $(x :: 'a::trans-reals) + 0 = x$   
**by** (*subst A2, simp*)

**lemma** *multiplicative-identity-right*[*simp*]:  $(x :: 'a::trans-reals) * 1 = x$   
**by** (*subst A13, simp*)

**lemma** *additive-nullity-right*[*simp*]:  $(x :: 'a::trans-reals) + \Phi = \Phi$   
**by** (*subst A2, simp*)

**lemma** *additive-infinity-right*:  $\llbracket x \neq -\infty; x \neq \Phi \rrbracket \implies (x :: 'a::trans-reals) + \infty = \infty$   
**by** (*subst A2, subst A5, auto*)

**lemma** *minus-minus*[*simp*]:  $(x :: 'a::trans-reals) - - y = x + y$   
**by** (*simp add: A6*)

**lemma** *zero-mult-infinity*[*simp*]:  
 $0 * (\infty :: 'a::trans-reals) = \Phi$   
**by** (*subst A13, rule A16*)

**lemma** *nullity-minus-left*[*simp*]:  $\Phi - (x :: 'a::trans-reals) = \Phi$   
**by** (*simp add: A6*)

**lemma** *nullity-minus-right*[*simp*]:  $(x :: 'a::trans-reals) - \Phi = \Phi$   
**by** (*simp add: A6*)

**lemma** *zero-minus-eq-uminus*[*simp*]:  $(0 :: 'a::trans-reals) - x = -x$   
**by** (*simp add: A6*)

**lemma** *uminus-eq-uminus*[*simp*]:  $(-(x :: 'a::trans-reals) = -y) = (x = y)$   
**by** (*safe, drule-tac f= $\lambda x. - x$  in arg-cong, simp*)

**lemma** *uminus-add-eq-minus*:  $-x + y = y - (x :: 'a::trans-reals)$   
**by** (*subst A2, simp add: A6*)

**lemma** *uminus-minus-commute*:  $-x - y = - y - (x :: 'a::trans-reals)$   
**by** (*simp add: A6 A2*)

**lemma** *x-add-y-minus-y*:  
 $\llbracket y \neq \Phi; y \neq \infty; y \neq -\infty \rrbracket \implies x + y - y = (x :: 'a::trans-reals)$   
**apply** (*subst A6, subst A1[THEN sym]*)  
**by** (*simp add: A6[THEN sym] A8*)

**lemma** *uminus-x-add-x*:  $\llbracket x \neq \infty; x \neq -\infty; x \neq \Phi \rrbracket \implies -x + x = (0 :: 'a :: \text{trans-reals})$

**by** (*subst* *A2*, *simp add*: *A6[THEN sym]* *A8*)

**lemma** *x-minus-y-add-y*:

$\llbracket y \neq \Phi; y \neq \infty; y \neq -\infty \rrbracket \implies x - y + y = (x :: 'a :: \text{trans-reals})$

**apply** (*subst* *A6*, *subst* *A1[THEN sym]*)

**by** (*simp add*: *uminus-x-add-x*)

**lemma** *uminus-eq-nullity-iff[simp]*:  $\llbracket x :: 'a :: \text{trans-reals} \rrbracket. (-x = \Phi) = (x = \Phi)$

**by** (*auto*, *drule-tac*  $f = \lambda x. -x$  **in** *arg-cong*, *simp*)

**lemma** *uminus-eq-infinity-iff[simp]*:  $\llbracket x :: 'a :: \text{trans-reals} \rrbracket. (-x = \infty) = (x = -\infty)$

**by** (*auto*, *drule-tac*  $f = \lambda x. -x$  **in** *arg-cong*, *simp*)

**lemma** *infinity-minus*:  $\llbracket x \neq \Phi; x \neq \infty \rrbracket \implies \infty - x = (\infty :: 'a :: \text{trans-reals})$

**by** (*subst* *A6*, *subst* *A5*, *simp-all*)

## 2.2 Distinctness of six basic constants: 0, nullity, +1,-1, plus/minus infinity

**lemma** *not-zero-less-zero[simp]*:  $\neg 0 < (0 :: 'a :: \text{trans-reals})$

**by** (*cut-tac*  $a = 0 :: 'a$  **in** *A28*, *auto*)

**lemma** *zero-noteq-nullity[simp]*:  $(0 :: 'a :: \text{trans-reals}) \neq \Phi$

**by** (*cut-tac*  $a = \Phi :: 'a$  **in** *A28*, *auto*)

**lemmas** *nullity-noteq-zero[simp]* = *zero-noteq-nullity[THEN not-sym]*

**lemma** *zero-noteq-infinity[simp]*:  $(0 :: 'a :: \text{trans-reals}) \neq \infty$

**by** (*cut-tac*  $a = (\infty :: 'a)$  **in** *A28*, *simp*)

**lemmas** *infinity-noteq-zero[simp]* = *zero-noteq-infinity[THEN not-sym]*

**lemma** *nullity-not-less-zero[simp]*:  $\neg \Phi < (0 :: 'a :: \text{trans-reals})$

**by** (*cut-tac*  $a = (\Phi :: 'a)$  **in** *A28*, *simp*)

**lemma** *zero-not-less-nullity[simp]*:  $\neg 0 < (\Phi :: 'a :: \text{trans-reals})$

**by** (*cut-tac*  $a = (\Phi :: 'a)$  **in** *A28*, *simp*)

**lemma** *infinity-noteq-nullity[simp]*:  $(\infty :: 'a :: \text{trans-reals}) \neq \Phi$

**by** (*rule notI*, *drule-tac*  $f = \lambda x. 0 < x$  **in** *arg-cong*, *simp*)

**lemmas** *nullity-noteq-infinity[simp]* = *infinity-noteq-nullity[THEN not-sym]*

**lemma** *zero-less-one[simp]*:  $(0 :: 'a :: \text{trans-reals}) < 1$

**by** (*simp add*: *A23[THEN sym]*)

**lemma** *not-one-less-zero*[simp]:  $\neg (1 < (0 :: 'a :: \text{trans-reals}))$   
**by** (*cut-tac*  $a = (1 :: 'a)$  **in** *A28*, *auto*)

**lemma** *zero-noteq-one*[simp]:  $(0 :: 'a :: \text{trans-reals}) \neq 1$   
**by** (*cut-tac*  $a = (1 :: 'a)$  **in** *A28*, *auto*)

**lemmas** *one-noteq-zero*[simp] = *zero-noteq-one* [THEN *not-sym*]

**lemma** *one-noteq-infinity*[simp]:  $(1 :: 'a :: \text{trans-reals}) \neq \infty$   
**apply** (*rule classical*, *simp*)  
**by** (*drule-tac*  $f = \lambda y. y * 0$  **in** *arg-cong*, *simp*)

**lemmas** *infinity-noteq-one*[simp] = *one-noteq-infinity* [THEN *not-sym*]

**lemma** *nullity-noteq-one*[simp]:  $\Phi \neq (1 :: 'a :: \text{trans-reals})$   
**by** (*rule notI*, *drule-tac*  $f = \lambda x. 0 < x$  **in** *arg-cong*, *simp*)

**lemmas** *one-not-nullity* [simp] = *nullity-noteq-one* [THEN *not-sym*]

**lemma** *minus-one-less-zero*[simp]:  $(- 1 :: 'a :: \text{trans-reals}) < 0$   
**by** (*subst* *A26* [THEN *sym*], *simp*)

**lemma** *minus-one-mult-infinity*[simp]:  $(- 1 :: 'a :: \text{trans-reals}) * \infty = - \infty$   
**by** (*subst* *A13*, *rule* *A24* [THEN *iffD2*], *rule* *minus-one-less-zero*)

**lemma** *zero-noteq-minus-one*[simp]:  $(0 :: 'a :: \text{trans-reals}) \neq - 1$   
**by** (*cut-tac*  $a = - (1 :: 'a)$  **in** *A28*, *auto*)

**lemmas** *minus-one-noteq-zero*[simp] = *zero-noteq-minus-one* [THEN *not-sym*]

**lemma** *not-one-less-minus-one*[simp]:  $\neg ((0 :: 'a :: \text{trans-reals}) < - 1)$   
**by** (*cut-tac*  $a = - (1 :: 'a)$  **in** *A28*, *auto*)

**lemma** *one-noteq-minus-one*[simp]:  $(1 :: 'a :: \text{trans-reals}) \neq - 1$   
**by** (*rule notI*, *drule-tac*  $f = \lambda x. x < 0$  **in** *arg-cong*, *simp*)

**lemmas** *minus-one-noteq-one*[simp] = *one-noteq-minus-one* [THEN *not-sym*]

**lemma** *infinity-noteq-minus-one*[simp]:  $\infty \neq (- 1 :: 'a :: \text{trans-reals})$   
**by** (*rule notI*, *drule-tac*  $f = \lambda x. 0 < x$  **in** *arg-cong*, *simp*)

**lemmas** *minus-one-not-infinity* [simp] = *infinity-noteq-minus-one* [THEN *not-sym*]

**lemma** *nullity-noteq-minus-one*[simp]:  $\Phi \neq (- 1 :: 'a :: \text{trans-reals})$   
**by** (*rule notI*, *drule-tac*  $f = \lambda x. x < 0$  **in** *arg-cong*, *simp*)

**lemmas** *minus-one-not-nullity* [simp] = *nullity-noteq-minus-one* [THEN *not-sym*]

**lemma** *zero-noteq-minus-infinity*[simp]:  $0 \neq (-\infty :: 'a::\text{trans-reals})$   
**apply** (cut-tac a=- ( $\infty :: 'a$ ) **in** A28, simp,safe)  
**apply** (erule notE)  
**apply** (rule A26[THEN iffD1])  
**by** (subst A6, subst A7, simp)

**lemmas** *minus-infinity-noteq-zero*[simp] = *zero-noteq-minus-infinity*[THEN not-sym]

**lemma** *uminus-zero*[simp]:  $-(0 :: 'a::\text{trans-reals}) = 0$   
**by** (cut-tac a=0::'a **in** A8, simp-all add: A6)

**lemma** *uminus-eq-zero-iff*[simp]:  $!! x::'a::\text{trans-reals}. (-x = 0) = (x = 0)$   
**by** (auto, drule-tac f= $\lambda x. -x$  **in** arg-cong, simp)

**lemma** *minus-infinity-less-zero*[simp]:  $(-\infty :: 'a::\text{trans-reals}) < 0$   
**by** (rule A26 [THEN iffD1], simp)

**lemma** *minus-infinity-mult-infinity*[simp]:  $(-\infty :: 'a::\text{trans-reals}) * \infty = -\infty$   
**by** (subst A13, rule A24[THEN iffD2], rule minus-infinity-less-zero)

**lemma** *not-zero-less-minus-infinity*[simp]:  $\neg 0 < (-\infty :: 'a::\text{trans-reals})$   
**by** (cut-tac a=- ( $\infty :: 'a$ ) **in** A28, auto)

**lemma** *one-noteq-minus-infinity*[simp]:  $1 \neq (-\infty :: 'a::\text{trans-reals})$   
**apply** (rule notI, drule-tac f= $\lambda x. -x$  **in** arg-cong)  
**by** (simp only: A7 minus-one-not-infinity)

**lemmas** *minus-infinity-noteq-one*[simp] = *one-noteq-minus-infinity* [THEN not-sym]

**lemma** *infinity-noteq-minus-infinity*[simp]:  $\infty \neq (-\infty :: 'a::\text{trans-reals})$   
**by** (rule notI, drule-tac f= $\lambda x. 0 < x$  **in** arg-cong, simp)

**lemmas** *minus-infinity-noteq-infinity*[simp]  
 = *infinity-noteq-minus-infinity*[THEN not-sym]

**lemma** *nullity-noteq-minus-infinity*[simp]:  $\Phi \neq (-\infty :: 'a::\text{trans-reals})$   
**by** (rule notI, drule-tac f= $\lambda x. -x$  **in** arg-cong, simp)

**lemmas** *minus-infinity-noteq-nullity*[simp]  
 = *nullity-noteq-minus-infinity*[THEN not-sym]

**lemma** *one-minus-one-eq-zero*[simp]:  $(1::'a::\text{trans-reals}) - 1 = 0$   
**by** (subst A8, auto)

**lemma** *less-irreflexive*[simp]:  $\neg x < (x::'a::\text{trans-reals})$   
**apply** (subst A26[THEN sym])



```

apply (case-tac  $x = \infty$ , simp)
apply (case-tac  $x = -\infty$ , simp add: A2 A6[THEN sym])
apply (case-tac  $x = \Phi$ , simp add: A6)
by (subst A8, auto)

```

TODO: less\_transitive

```

lemma not-less-zero-and-zero-less:
   $\llbracket x < (0 :: 'a::trans-reals); 0 < x \rrbracket \implies P$ 
by (cut-tac a=( $x :: 'a$ ) in A28, auto)

```

```

lemma less-zero-imp-not-zero:  $x < (0 :: 'a::trans-reals) \implies x \neq 0$ 
by auto

```

```

lemma zero-less-imp-not-zero:  $(0 :: 'a::trans-reals) < x \implies x \neq 0$ 
by auto

```

## 2.3 sign equivalences

```

lemma sgn-negative-iff:
   $!! a :: 'a::trans-reals. (sgn\ a = -\ 1) = (a < 0)$ 
apply (unfold trans-sgn, auto)
by (erule not-less-zero-and-zero-less, auto)

```

```

lemma sgn-positive-iff:
   $!! a :: 'a::trans-reals. (sgn\ a = 1) = (0 < a)$ 
by (unfold trans-sgn, auto)

```

```

lemma sgn-zero[simp]:  $sgn\ (0 :: 'a::trans-reals) = 0$ 
by (unfold trans-sgn, simp)

```

```

lemma sgn-zero-iff:  $(sgn\ (x :: 'a::trans-reals) = 0) = (x = 0)$ 
by (unfold trans-sgn, simp)

```

```

lemma sgn-nullity[simp]:  $sgn\ (\Phi :: 'a::trans-reals) = \Phi$ 
by (unfold trans-sgn, simp)

```

```

lemma sgn-nullity-iff:  $(sgn\ (x :: 'a::trans-reals) = \Phi) = (x = \Phi)$ 
apply (unfold trans-sgn, simp)
by (cut-tac a=( $x :: 'a$ ) in A28, auto)

```

## 2.4 Algebraic stuff and inequalities...

```

lemma infinity-add-infinity[simp]:  $(\infty :: 'a::trans-reals) + \infty = \infty$ 
by (simp add: A5)

```

```

lemma minus-infinity-minus-infinity[simp]:  $-(\infty :: 'a::trans-reals) - \infty = -\infty$ 
by (simp add: A10)

```

```

lemma minus-infinity-add-infinity[simp]:  $-(\infty :: 'a::trans-reals) + \infty = \Phi$ 

```

by (subst A2, simp add: A6[THEN sym])

**lemma not-nullity-less[simp]:**  $\neg \Phi < (x :: 'a :: \text{trans-reals})$   
 apply (subst A26[THEN sym])  
 apply (simp add: A23[THEN sym] A6)  
 by (subst A13, simp)

**lemma not-less-nullity[simp]:**  $\neg x < (\Phi :: 'a :: \text{trans-reals})$   
 apply (subst A26[THEN sym])  
 apply (simp add: A23[THEN sym] A6)  
 by (subst A13, simp)

**lemma not-nullity-le-infinity:**  $x \neq \Phi \implies x < \infty \mid x = (\infty :: 'a :: \text{trans-reals})$   
 by (subst A26[THEN sym], clarsimp simp: infinity-minus)

**lemma less-infinity-iff:**  $(x < \infty) = (x \neq (\infty :: 'a :: \text{trans-reals}) \wedge x \neq \Phi)$   
 by (auto dest: not-nullity-le-infinity)

**lemma not-nullity-minus-infinity-le:**  $x \neq \Phi \implies -\infty < x \mid x = (-\infty :: 'a :: \text{trans-reals})$   
 by (subst A26[THEN sym], clarsimp simp: infinity-minus additive-infinity-right)

**lemma minus-infinity-less-iff:**  $(-\infty < x) = (x \neq (-\infty :: 'a :: \text{trans-reals}) \wedge x \neq \Phi)$   
 by (auto dest: not-nullity-minus-infinity-le)

**lemma infinity-add-eq-nullity:**  
 $(\infty + x = \Phi) = (x = -(\infty :: 'a :: \text{trans-reals}) \vee x = \Phi)$   
 apply (safe, simp-all add: A6[THEN sym])  
 apply (case-tac x= $\infty$ , simp)  
 apply (drule-tac f= $\lambda a. a - x$  in arg-cong)  
 apply (erule-tac Q=?x = ?y in contrapos-pp)  
 apply (simp add: A1[THEN sym] A6)  
 by (simp add: A6[THEN sym] A8)

**lemma infinity-minus-eq-nullity:**  
 $(x - \infty = \Phi) = (x = (\infty :: 'a :: \text{trans-reals}) \vee x = \Phi)$   
 apply (safe, simp-all add: A6[THEN sym])  
 apply (case-tac x= $-\infty$ , simp)  
 apply (drule-tac f= $\lambda a. -x + a$  in arg-cong)  
 apply (erule-tac Q=?x = ?y in contrapos-pp)  
 by (simp add: A1 A6 uminus-x-add-x)

**lemma add-eq-nullity-iff:**  
 $(x + y = (\Phi :: 'a :: \text{trans-reals})) =$   
 $(x = \Phi \vee y = \Phi \vee (x = \infty \wedge y = -\infty) \vee (x = -\infty \wedge y = \infty))$   
 apply auto  
 apply (simp-all add: A6[THEN sym])

apply (drule-tac f= $\lambda a. -x + a$  in arg-cong)

```

apply (erule-tac  $Q = ?x = ?y$  in contrapos-pp)
apply (simp add: A1 uminus-x-add-x)

apply (case-tac  $x = \infty$ , simp add: infinity-add-eq-nullity)
apply (drule-tac  $f = \lambda a. - x + a$  in arg-cong)
apply (erule-tac  $Q = ?x = ?y$  in contrapos-pp)
apply (simp add: A1 uminus-x-add-x)

apply (case-tac  $y = -\infty$ , simp add: A6[THEN sym] infinity-minus-eq-nullity)
apply (drule-tac  $f = \lambda a. a - y$  in arg-cong)
apply (erule-tac  $Q = ?x = ?y$  in contrapos-pp)
apply (simp add: x-add-y-minus-y)

apply (drule-tac  $f = \lambda a. a - y$  in arg-cong)
apply (erule-tac  $Q = ?x = ?y$  in contrapos-pp)
by (simp add: x-add-y-minus-y)

lemma minus-nullity-eq-iff:
   $(x - y = (\Phi :: 'a :: \text{trans-reals})) =$ 
     $(x = \Phi \vee y = \Phi \vee (x = \infty \wedge y = \infty) \vee (x = -\infty \wedge y = -\infty))$ 

by (simp add: A6 add-eq-nullity-iff)

lemma add-infinity:
   $x + (\infty :: 'a :: \text{trans-reals}) = (\text{if } x = \Phi \mid x = -\infty \text{ then } \Phi \text{ else } \infty)$ 
apply (case-tac  $x = \Phi$ , simp)
apply (case-tac  $x = -\infty$ , subst A2, simp add: A6[THEN sym])
by (subst A2, simp add: A5)

lemma add-infinity-not-eq[simp]:
   $(x :: 'a :: \text{trans-reals}) + \infty \neq 0 \wedge (x :: 'a :: \text{trans-reals}) + \infty \neq -\infty$ 
by (simp add: add-infinity)

lemma subtract-infinity:
   $x - (\infty :: 'a :: \text{trans-reals}) = (\text{if } x = \Phi \mid x = \infty \text{ then } \Phi \text{ else } -\infty)$ 
apply (case-tac  $x = \Phi$ , simp)
apply (case-tac  $x = \infty$ , simp)
by (simp add: A10)

lemma subtract-infinity-not-eq[simp]:
   $(x :: 'a :: \text{trans-reals}) - \infty \neq 0 \wedge (x :: 'a :: \text{trans-reals}) - \infty \neq \infty$ 
by (simp add: subtract-infinity)

lemma minus-nullD:  $x - y = (0 :: 'a :: \text{trans-reals}) \implies x = y$ 
apply (case-tac  $y = \Phi$ , simp)
apply (case-tac  $y = \infty$ , simp)
apply (case-tac  $y = -\infty$ , simp)
apply (drule-tac  $f = \lambda a. a + y$  in arg-cong)
apply (erule-tac  $Q = ?x = ?y$  in contrapos-pp)

```

**by** (*simp add*: A6 A1[*THEN sym*] *uminus-x-add-x*)

**lemma** *add-cancel-right*:  $\llbracket x + a = y + a; a \neq \Phi; a \neq \infty; a \neq -\infty \rrbracket$   
 $\implies x = (y :: 'a::trans-reals)$   
**apply** (*drule-tac*  $f = \lambda x. x - a$  **in** *arg-cong*)  
**by** (*simp add*: *x-add-y-minus-y*)

**lemma** *add-cancel-left*:  $\llbracket a + x = a + y; a \neq \Phi; a \neq \infty; a \neq -\infty \rrbracket$   
 $\implies x = (y :: 'a::trans-reals)$   
**apply** (*drule-tac*  $f = \lambda x. x - a$  **in** *arg-cong*)  
**apply** (*simp add*: A6)  
**apply** (*erule-tac*  $Q=?a=?b$  **in** *contrapos-pp*)  
**apply** (*subst add-commute*)  
**apply** (*subst add-commute* [**where**  $a=a$ ])  
**apply** (*simp add*: *add-assoc*)  
**by** (*simp add*: A6[*THEN sym*] A8)

**lemma** *add-eq-infinity-iff*:  
 $(x + y = (\infty :: 'a::trans-reals)) =$   
 $((x = \infty \wedge y \neq \Phi \wedge y \neq -\infty) \vee (y = \infty \wedge x \neq \Phi \wedge x \neq -\infty))$   
**apply** *safe*  
**apply** (*simp-all add*: A6[*THEN sym*] A5 *uminus-add-eq-minus*) **defer**  
**apply** (*subst* A2, *simp add*: A5)  
**apply** (*case-tac*  $y=\Phi$ , *simp*)  
**apply** (*case-tac*  $y=-\infty$ , *simp add*: A6[*THEN sym*])  
**apply** (*rule-tac*  $a=y$  **in** *add-cancel-right*)  
**by** (*simp add*: A5, *assumption+*)

**lemma** *add-eq-minus-infinity-iff*:  
 $(x + y = (-\infty :: 'a::trans-reals)) =$   
 $((x = -\infty \wedge y \neq \Phi \wedge y \neq \infty) \vee (y = -\infty \wedge x \neq \Phi \wedge x \neq \infty))$   
**apply** *safe*  
**apply** (*simp-all add*: A6[*THEN sym*] A10 *uminus-add-eq-minus*) **defer**  
**apply** (*erule contrapos-pp*, *subst* A2, *simp*)  
**apply** (*case-tac*  $y=\Phi$ , *simp*)  
**apply** (*case-tac*  $y=\infty$ , *simp add*: A6[*THEN sym*])  
**apply** (*rule-tac*  $a=y$  **in** *add-cancel-right*)  
**by** (*simp add*: *uminus-add-eq-minus* A10, *assumption+*)

**lemma** *reals-add*:  $!! x::'a::trans-reals.$   
 $\llbracket x \neq \Phi; x \neq \infty; x \neq -\infty; y \neq \Phi; y \neq \infty; y \neq -\infty \rrbracket$   
 $\implies (x + y) \neq \Phi \wedge (x + y) \neq \infty \wedge (x + y) \neq -\infty$   
**apply** *safe*  
**apply** (*simp add*: *add-eq-nullity-iff*)  
**apply** (*simp add*: *add-eq-infinity-iff*)  
**by** (*simp add*: *add-eq-minus-infinity-iff*)

## 2.5 Closure of reals under addition

**lemma** *reals-add-closed*:  $\llbracket x \in \text{reals}; y \in \text{reals} \rrbracket \implies x + y \in \text{reals}$

**by** (*unfold reals-def*, *simp add: reals-add*)

Distributivity of uminus over addition

**lemma** *uminus-distrib-add*:  $-(x + y) = -x - (y :: 'a::\text{trans-reals})$

**apply** (*case-tac x=Φ*, *simp*)  
**apply** (*tactic ALLGOALS(case-tac y=Φ)*, *simp-all*)  
**apply** (*case-tac x=∞*, *simp*)  
**apply** (*tactic ALLGOALS(case-tac y=∞)*, *simp-all*)  
**apply** (*tactic ALLGOALS(case-tac x=-∞)*, *simp-all*)  
**apply** (*tactic ALLGOALS(case-tac y=-∞)*)  
**apply** (*simp-all add: A6[THEN sym] uminus-add-eq-minus infinity-minus A10*)

**apply** (*simp add: A5, subst uminus-minus-commute*)  
**apply** (*simp add: A10*)  
**apply** (*subst A2, simp add: A5*)

**apply** (*rule-tac a = x in add-cancel-left*)  
**apply** (*simp add: A6 A1*)  
**apply** (*simp add: A6[THEN sym] A8*)

**apply** (*rule-tac a = y in add-cancel-left*)  
**apply** (*simp add: A6[THEN sym] A8*)  
**apply** (*simp add: A6*)  
**apply** (*subst A1[where a = y]*)  
**apply** (*simp add: A6[THEN sym]*)  
**apply** (*subst A2 [where a=y]*)  
**apply** (*rule A8*)  
**by** (*simp-all add: reals-add*)

**lemma** *uminus-distrib-minus*:  $-(x - y) = -x + (y :: 'a::\text{trans-reals})$

**by** (*simp add: A6 uminus-distrib-add*)

**lemma** *ordering-opp*:  $(x - y < 0) = (x < (y :: 'a::\text{trans-reals}))$

**apply** (*subst A26 [THEN sym]*)  
**apply** (*simp add: A6 uminus-distrib-add uminus-add-eq-minus*)  
**by** (*simp add: A26 A6[THEN sym]*)

Trichotomy

**lemma** *trichotomy*:

$\llbracket x \neq \Phi; y \neq (\Phi :: 'a::\text{trans-reals}) \rrbracket \implies (x < y) \mid (x = y) \mid (y < x)$   
**apply** (*cut-tac a=((x-y)::'a) in A28, auto*)  
**apply** (*simp-all add: A26*)  
**apply** (*simp add: minus-nullity-eq-iff, safe*)  
**apply** (*erule minus-nullD*)  
**by** (*simp add: ordering-opp*)

**lemma** *reals-uminus*: !!  $x :: 'a :: \text{trans-reals}$ .  
 $\llbracket x \neq \Phi; x \neq \infty; x \neq -\infty \rrbracket \implies -x \neq \Phi \wedge -x \neq \infty \wedge -x \neq -\infty$   
**by** (*safe*, *simp-all*)

## 2.6 Closure of reals under uminus

**lemma** *reals-uminus-closed*:  $x \in \text{reals} \implies -x \in \text{reals}$   
**by** (*unfold reals-def*, *simp add: reals-add*)

Closure of positives transreals under addition

**lemma** *zero-less-add*:  
!!  $a :: 'a :: \text{trans-reals}$ .  $\llbracket 0 < a; 0 < b \rrbracket \implies 0 < a + b$   
**apply** (*subst A23[THEN sym]*)  
**apply** (*subst A29, auto*)  
**apply** (*simp add: sgn-positive-iff[THEN iffD2]*)  
**by** (*simp add: A23[THEN iffD2]*)

**lemma** *reals-interval*:  $\text{reals} = \{x. -\infty < x \wedge x < \infty\}$   
**by** (*auto simp: reals-def less-infinity-iff minus-infinity-less-iff*)

**lemma** *reals-cases*:  $x \in \text{reals} \implies (x < 0) \mid x = 0 \mid (0 < x)$   
**apply** (*unfold reals-def, clarsimp*)  
**by** (*cut-tac A28 [where a=x], simp*)

**lemma** *not-infinity-less[simp]*:  $\neg (\infty < (x :: 'a :: \text{trans-reals}))$   
**by** (*subst A26[THEN sym], simp add: subtract-infinity*)

**lemma** *not-less-minus-infinity[simp]*:  $\neg ((x :: 'a :: \text{trans-reals}) < -\infty)$   
**apply** (*subst A26[THEN sym]*)  
**apply** (*subst uminus-minus-commute*)  
**by** (*simp add: subtract-infinity*)

Left distributivity of multiplication over addition

**lemma** *distrib-left*:  
 $\neg ((c = \infty \vee c = -\infty) \wedge \text{sgn } a \neq \text{sgn } b \wedge (a + b \notin \{0, \Phi\}))$   
 $\implies (a + b) * c = (a * c) + (b * (c :: 'a :: \text{trans-reals}))$   
**apply** (*subst A13[where a=a+b]*)  
**apply** (*subst A13[where a=a]*)  
**apply** (*subst A13[where a=b]*)  
**by** (*simp add: A29*)

**lemma** *zero-mult-not-less-zero*:  $\neg ((0 :: 'a :: \text{trans-reals}) * x < 0)$   
**by** (*simp add: A24[THEN sym] A12*)

**lemma** *not-zero-less-zero-mult*:  $\neg (0 < (0 :: 'a :: \text{trans-reals}) * x)$   
**by** (*simp add: A23[THEN sym] A12*)

**lemma** *zero-mult-eq-zero-or-nullity*:  $(0::'a::\text{trans-reals}) * x = 0 \vee 0 * x = \Phi$   
**apply** (*cut-tac* *A28* [**where**  $a=0 * x$ ])  
**by** (*auto simp*: *zero-mult-not-less-zero not-zero-less-zero-mult*)

**lemma** *zero-mult-minus-one*:  $(0::'a::\text{trans-reals}) * - 1 = 0$   
**apply** (*cut-tac* *zero-mult-eq-zero-or-nullity* [**where**  $x=- 1::'a$ ], *clarsimp*)  
**apply** (*drule-tac*  $f=\lambda x. x + 0 * (1 + 1)$  **in** *arg-cong*)  
**by** (*simp add*: *A29[THEN sym] A1 uminus-add-eq-minus A8*)

**lemma** *zero-mult-zero*:  $(0::'a::\text{trans-reals}) * 0 = 0$   
**apply** (*subgoal-tac*  $(0::'a::\text{trans-reals}) * (1 + - 1) = 0$ )  
**apply** (*simp add*: *A6 [THEN sym]*)  
**by** (*simp add*: *A29 zero-mult-minus-one*)

Multiplicative inverse

**lemma** *mult-inverse*:  $\llbracket a \neq 0; a: \text{reals} \rrbracket \implies a * \text{inverse } a = (1::'a::\text{trans-reals})$   
**by** (*unfold reals-def*, *clarsimp*, *drule A18*, *assumption+*, *simp add*: *A17*)

**lemma** *zero-mult-real-not-zero*:  $\llbracket (x::'a::\text{trans-reals}): \text{reals}; x \neq 0 \rrbracket \implies 0 * x = 0$   
**apply** (*cut-tac* *zero-mult-eq-zero-or-nullity* [**where**  $x=x$ ], *safe*)  
**apply** (*drule-tac*  $f=\lambda y. y * \text{inverse } x$  **in** *arg-cong*)  
**by** (*simp add*: *A12 [THEN sym] mult-inverse*)

Multiplication with zero

**lemma** *zero-mult-reals*:  $x : \text{reals} \implies 0 * x = (0::'a::\text{trans-reals}) \wedge x * 0 = 0$   
**apply** (*rule context-conjI*)  
**apply** (*case-tac*  $x=0$ )  
**apply** (*simp add*: *zero-mult-zero*)  
**apply** (*erule zero-mult-real-not-zero*, *assumption*)  
**by** (*subst A13*, *assumption*)

**lemmas** *zero-mult* = *zero-mult-reals* [*unfolded reals-def*, *simplified*]

**lemma** *mult-zero*:  $x \neq \Phi \wedge x \neq \infty \wedge x \neq -\infty \implies x * (0::'a::\text{trans-reals}) = 0$   
**apply** (*rule trans*) **defer**  
**apply** (*rule zero-mult* [*THEN conjunct1*], *assumption*)  
**by** (*rule A13*)

Multiplication with -1

**lemma** *minus-one-mult-uminus-reals*:  
 $x : \text{reals} \implies (- 1 ::'a::\text{trans-reals}) * x = -x$   
**apply** (*rule-tac*  $a=x$  **in** *add-cancel-right*, *unfold reals-def*, *safe*)  
**apply** (*simp add*: *uminus-add-eq-minus A8*)  
**apply** (*subst A14* [**where**  $a=x$ , *THEN sym*]) **back**  
**apply** (*subst A13*)  
**apply** (*subst A13*) **back**  
**apply** (*subst A29[THEN sym]*)  
**apply** *simp*

by (simp add: uminus-add-eq-minus A8 mult-zero)

**lemma** minus-one-mult-minus-one[simp]:  $(-1 :: 'a::trans-reals) * -1 = 1$   
 by (simp add: minus-one-mult-uminus-reals reals-def)

**lemma** minus-one-mult-uminus:  
 $(-1 :: 'a::trans-reals) * x = -x$   
 apply (case-tac x=Φ, simp add: A13)  
 apply (case-tac x=∞, simp)  
 apply (case-tac x=-∞, clarsimp)  
 apply (subst minus-one-mult-infinity[THEN sym])  
 apply (subst A12)  
 apply (subst minus-one-mult-minus-one, simp)  
 apply (rule minus-one-mult-uminus-reals)  
 by (simp add: reals-def)

**lemma** minus-one-mult-minus-infinity[simp]:  $(-1 :: 'a::trans-reals) * -∞ = ∞$   
 by (simp add: minus-one-mult-uminus)

**lemma** mult-nullity-right[simp]:  $(x :: 'a::trans-reals) * Φ = Φ$   
 by (subst A13, rule A15)

**lemma** zero-mult-minus-infinity[simp]:  $(0 :: 'a::trans-reals) * -∞ = Φ$   
 apply (subst minus-one-mult-infinity[THEN sym])  
 apply (subst A12)  
 by (simp add: zero-mult)

**lemma** minus-infinity-mult-zero[simp]:  $-∞ * (0 :: 'a::trans-reals) = Φ$   
 by (subst A13, rule zero-mult-minus-infinity)

**lemma** uminus-mult-left:  $-((x :: 'a::trans-reals) * y) = (-x) * y$   
 apply (subst minus-one-mult-uminus[THEN sym])  
 apply (subst A12)  
 by (simp add: minus-one-mult-uminus)

**lemma** uminus-mult-right:  $-((x :: 'a::trans-reals) * y) = x * (-y)$   
 apply (subst minus-one-mult-uminus[THEN sym])  
 apply (subst A13)  
 apply (subst A12 [THEN sym])  
 apply (subst A13 [where a=y])  
 by (simp add: minus-one-mult-uminus)

**lemma** uminus-mult-uminus [simp]:  $-(x :: 'a::trans-reals) * (-y) = x * y$   
 apply (subst uminus-mult-left[THEN sym])  
 by (simp add: uminus-mult-right)

**lemma** zero-less-uminus-iff-less-zero:  $(0 < -x) = ((x :: 'a::trans-reals) < 0)$   
 apply (simp add: A23[THEN sym] A24[THEN sym], auto)  
 apply (drule-tac f =λ x. -x in arg-cong)



```

apply (simp add: uminus-mult-right)
apply (drule-tac f =  $\lambda x. -x$  in arg-cong)
by (simp add: uminus-mult-right)

```

```

lemma uminus-less-zero-iff-zero-less:  $(-x < 0) = (0 < (x::'a::trans-reals))$ 
apply (simp add: A23[THEN sym] A24[THEN sym], auto)
apply (drule-tac f =  $\lambda x. -x$  in arg-cong)
apply (simp add: uminus-mult-right)
apply (drule-tac f =  $\lambda x. -x$  in arg-cong)
by (simp add: uminus-mult-right)

```

```

lemma less-not-sym:  $(x::'a::trans-reals) < y \implies \neg (y < x)$ 
apply (rule notI, drule A26[THEN iffD2], drule A26[THEN iffD2])
apply (drule uminus-less-zero-iff-zero-less[THEN iffD2])
apply (simp add: uminus-distrib-minus uminus-add-eq-minus)
by (erule not-less-zero-and-zero-less, assumption)

```

Closure of positive transreals under multiplication

```

lemma mult-gt-zero-gt-zero:
  !! a :: 'a::trans-reals.  $\llbracket 0 < a; 0 < b \rrbracket \implies 0 < a * b$ 
apply (subst A23[THEN sym])
apply (subst A12)
apply (subst A23[THEN iffD2], assumption)
by (erule A23[THEN iffD2])

```

Mixed-sign multiplication

```

lemma mult-gt-zero-less-zero:
  !! a :: 'a::trans-reals.  $\llbracket 0 < a; b < 0 \rrbracket \implies a * b < 0$ 
apply (subst A24[THEN sym])
apply (subst A12)
apply (subst A23[THEN iffD2], assumption)
by (erule A24[THEN iffD2])

```

```

lemma mult-less-zero-gt-zero:
  !! a :: 'a::trans-reals.  $\llbracket a < 0; 0 < b \rrbracket \implies a * b < 0$ 
by (subst A13, erule mult-gt-zero-less-zero, assumption)

```

Multiplication of two negative transreals

```

lemma mult-less-zero-less-zero:
  !! a :: 'a::trans-reals.  $\llbracket a < 0; b < 0 \rrbracket \implies 0 < a * b$ 
apply (subst A23[THEN sym])
apply (subst A12)
apply (subst A24[THEN iffD2], assumption)
apply (subst minus-one-mult-infinity [THEN sym])
apply (subst A12[THEN sym])
by (subst A24 [THEN iffD2], auto)

```

```

lemma infinity-mult:
   $\infty * (x::'a::trans-reals) =$ 

```

(if  $x < 0$  then  $-\infty$  else if  $0 < x$  then  $\infty$  else  $\Phi$ )  
**apply** (auto simp add: A23 A24)  
**by** (cut-tac A28 [where a=x], auto)

**lemma** mult-infinity:  
 ( $x :: 'a :: \text{trans-reals}$ ) \*  $\infty$  =  
 (if  $x < 0$  then  $-\infty$  else if  $0 < x$  then  $\infty$  else  $\Phi$ )  
**by** (subst A13, rule infinity-mult)

**lemma** minus-infinity-mult:  
 $-\infty$  \* ( $x :: 'a :: \text{trans-reals}$ ) =  
 (if  $x < 0$  then  $\infty$  else if  $0 < x$  then  $-\infty$  else  $\Phi$ )  
**apply** (subst minus-one-mult-infinity [THEN sym])  
**apply** (subst A12 [THEN sym])  
**apply** (subst infinity-mult)  
**by** (auto simp: minus-one-mult-uminus)

**lemma** mult-minus-infinity:  
 ( $x :: 'a :: \text{trans-reals}$ ) \*  $-\infty$  =  
 (if  $x < 0$  then  $\infty$  else if  $0 < x$  then  $-\infty$  else  $\Phi$ )  
**by** (subst A13, rule minus-infinity-mult)

Some inverse rules

**lemma** inverse-infinity[simp]:  $\text{inverse } (\infty) = (0 :: 'a :: \text{trans-reals})$   
**apply** (subst A20 [THEN sym])  
**apply** (rule A19)  
**by** (cut-tac a=( $\infty :: 'a$ ) in A28, auto)

**lemma** inverse-nullity-iff:  
 !!  $x :: 'a :: \text{trans-reals}$ . ( $\text{inverse } x = \Phi$ ) = ( $x = \Phi$ )  
**apply** (safe, simp-all)  
**apply** (frule-tac f=inverse in arg-cong)  
**apply** (case-tac x= $-\infty$ , simp)  
**by** (thin-tac inverse  $x = \Phi$ , simp add: A19)

**lemma** inverse-noteq-zero:  
 !!  $x :: 'a :: \text{trans-reals}$ .  $\llbracket x \neq \infty; x \neq -\infty \rrbracket \implies \text{inverse } x \neq 0$   
**apply** (case-tac x= $\Phi$ , simp)  
**apply** (subst inverse-infinity [THEN sym])  
**apply** (rule notI)  
**by** (drule-tac f=inverse in arg-cong, simp add: A19)

**lemma** inverse-zero-iff[simp]:  
 !!  $x :: 'a :: \text{trans-reals}$ . ( $\text{inverse } x = 0$ ) = ( $x = \infty \mid x = -\infty$ )  
**by** (auto, rule classical, cut-tac x=x in inverse-noteq-zero, auto)

**lemma** inverse-infinity-iff[simp]:  
 !!  $x :: 'a :: \text{trans-reals}$ . ( $\text{inverse } x = \infty$ ) = ( $x = 0$ )  
**apply** (case-tac x= $\Phi$ , simp)

```

apply (case-tac  $x=\infty$ , simp)
apply (case-tac  $x=-\infty$ , simp)
apply (case-tac  $x=0$ , auto)
apply (drule-tac  $f = \lambda a. x * a$  in arg-cong)
apply (simp add: mult-inverse reals-def mult-infinity)
apply (case-tac  $x < 0$ , simp)
by (case-tac  $0 < x$ , simp-all)

```

```

lemma inverse-minus-infinity-iff[simp]: !!  $x::'a::trans-reals$ . ( $inverse\ x \neq -\infty$ )
apply (case-tac  $x=\Phi$ , simp)
apply (case-tac  $x=\infty$ , simp)
apply (case-tac  $x=-\infty$ , simp)
apply (case-tac  $x=0$ , auto)
apply (drule-tac  $f = \lambda a. x * a$  in arg-cong)
apply (simp add: mult-inverse reals-def mult-minus-infinity)
apply (case-tac  $x < 0$ , simp)
by (case-tac  $0 < x$ , simp-all)

```

## 2.7 Closure of positive reals under inverse

```

lemma zero-less-inverse:
  !!  $x::'a::trans-reals$ . [ $0 < x$ ;  $x \neq \infty$ ]  $\implies 0 < inverse\ x$ 
apply (case-tac  $x=\Phi$ , simp)
apply (case-tac  $x=0$ , simp)
apply (case-tac  $x=-\infty$ , simp)
apply (cut-tac  $a=x$  in A18, assumption+)
apply (simp add: A17)
apply (subgoal-tac  $inverse\ x \neq \Phi$ ) defer
apply (simp add: inverse-nullity-iff)
apply (drule-tac trichotomy[where  $y = 0$ ]) back
apply (auto)
by (drule-tac mult-gt-zero-less-zero, assumption, simp)

```

```

lemma inverse-less-zero:
  !! ( $x::'a::trans-reals$ )  $< 0$ ;  $x \neq -\infty$   $\implies inverse\ x < 0$ 
apply (case-tac  $x=\Phi$ , simp)
apply (case-tac  $x=0$ , simp)
apply (case-tac  $x=\infty$ , simp)
apply (cut-tac  $a=x$  in A18, assumption+)
apply (simp add: A17)
apply (subgoal-tac  $inverse\ x \neq \Phi$ ) defer
apply (simp add: inverse-nullity-iff)
apply (cut-tac reals-cases [where  $x = inverse\ x$ ], auto)
apply (drule-tac mult-less-zero-gt-zero, assumption, simp)
by (auto simp: reals-def)

```

```

lemma inverse-less-zero-iff[simp]:
  ( $x::'a::trans-reals$ )  $\neq -\infty \implies (inverse\ x < 0) = (x < 0)$ 
apply (rule iffI)

```

**apply** (*drule-tac inverse-less-zero, simp*)  
**apply** (*simp add: A19*)  
**by** (*erule inverse-less-zero, assumption*)

**lemma** *zero-less-inverse-iff* [*simp*]:  
 $\llbracket (x::'a::\text{trans-reals}) \neq \infty; x \neq 0 \rrbracket \implies (0 < \text{inverse } x) = (0 < x)$   
**apply** (*case-tac x=-∞, simp*)  
**apply** (*rule iffI*)  
**apply** (*drule-tac zero-less-inverse, simp*)  
**apply** (*simp add: A19*)  
**by** (*erule zero-less-inverse, assumption*)

**lemma** *less-not-nullity*:  
 $(x::'a::\text{trans-reals}) < y \implies x \neq \Phi \wedge y \neq \Phi \wedge x \neq \infty \wedge y \neq -\infty$   
**by** (*auto*)

## 2.8 Closure of reals under multiplication

**lemma** *reals-mult-closed*:  $\llbracket x : \text{reals}; y : \text{reals} \rrbracket \implies x * y : \text{reals}$   
**apply** (*frule-tac x=x in reals-cases, frule-tac x=y in reals-cases, safe*)  
**apply** (*simp-all add: zero-mult-reals*)

**apply** (*frule mult-less-zero-less-zero, assumption*) **back**  
**apply** (*subst reals-def, clarsimp simp: less-not-nullity*)  
**apply** (*drule-tac f = λ a. a \* inverse y in arg-cong*)  
**apply** (*simp add: A12 [THEN sym]*)  
*mult-inverse less-zero-imp-not-zero infinity-mult reals-def*

**apply** (*frule mult-less-zero-gt-zero, assumption*)  
**apply** (*subst reals-def, clarsimp simp: less-not-nullity*)  
**apply** (*drule-tac f = λ a. a \* inverse y in arg-cong*)  
**apply** (*frule-tac y=y in less-not-sym*)  
**apply** (*simp add: A12 [THEN sym]*)  
*mult-inverse zero-less-imp-not-zero minus-infinity-mult reals-def*

**apply** (*frule mult-gt-zero-less-zero, assumption*)  
**apply** (*subst reals-def, clarsimp simp: less-not-nullity*)  
**apply** (*drule-tac f = λ a. a \* inverse y in arg-cong*)  
**apply** (*simp add: A12 [THEN sym]*)  
*mult-inverse less-zero-imp-not-zero minus-infinity-mult reals-def*

**apply** (*frule mult-gt-zero-gt-zero, assumption*) **back**  
**apply** (*subst reals-def, clarsimp simp: less-not-nullity*)  
**apply** (*drule-tac f = λ a. a \* inverse y in arg-cong*)  
**apply** (*frule-tac y=y in less-not-sym*)  
**by** (*simp add: A12 [THEN sym]*)  
*mult-inverse zero-less-imp-not-zero infinity-mult reals-def*

**end**

### 3 A Model for Transarithmetic

**theory** *TransNumberModel*

**imports** *TransNumberAxc* *Real*

**begin**

**datatype** *'a trans-number* = *P 'a* | *Infinity* | *MinusInfinity* | *Nullity*

**lemma** *inv-R-R[simp]*: *inv P (P x) = x*  
**by** (*rule inv-f-f, unfold inj-on-def, auto*)

**consts**

*primitive* :: *'a trans-number*  $\Rightarrow$  *bool*

**primrec**

*primitive* (*P x*) = *True*

*primitive Infinity* = *False*

*primitive MinusInfinity* = *False*

*primitive Nullity* = *False*

**lemma** *primitive-iff-exists*: *primitive x* = ( $\exists$  *r. x = P r*)  
**by** (*case-tac x, auto*)

**lemma** *primitiveD*: *primitive x*  $\Longrightarrow$   $\exists$  *r. x = P r*  
**by** (*simp add: primitive-iff-exists*)

**instance** *trans-number* :: (*zero*) *zero* ..

**instance** *trans-number* :: (*type*) *infinity* ..

**instance** *trans-number* :: (*type*) *nullity* ..

**instance** *trans-number* :: (*plus*) *plus* ..

**instance** *trans-number* :: (*minus*) *minus* ..

**defs** (**overloaded**)

*trans-number-zero-def*: *0 == P 0*

*trans-number-infinity-def*:  $\infty == \text{Infinity}$   
*trans-number-nullity-def*:  $\Phi == \text{Nullity}$

Note type annotations below, and overloaded symbol awkwardness

## 4 A model for axiomatic class `trans_add`

**primrec**

$P \ (x::'a::\{\text{plus}, \text{minus}\}) + y =$   
 $\quad (\text{if primitive } y \text{ then } P \ (x + \text{inv } P \ y) \text{ else}$   
 $\quad \text{if } y = \infty \quad \text{then } \infty \text{ else}$   
 $\quad \text{if } y = -\infty \quad \text{then } -\infty$   
 $\quad \text{else } \Phi)$   
 $\text{Infinity} + (y::'a::\{\text{plus}, \text{minus}\} \text{ trans-number})$   
 $\quad = (\text{if primitive } y \vee y = \infty \text{ then } \infty \text{ else } \Phi)$   
 $\text{MinusInfinity} + (y::'a::\{\text{plus}, \text{minus}\} \text{ trans-number})$   
 $\quad = (\text{if primitive } y \vee y = -\infty \text{ then } -\infty \text{ else } \Phi)$   
*Nullity-add-left*:  
 $\text{Nullity} + (y::'a::\{\text{plus}, \text{minus}\} \text{ trans-number}) = \Phi$

**primrec**

*uminus-P*:  $- P \ x = P \ (- \ (x))$   
*uminus-Infinity*:  $- \text{Infinity} = \text{MinusInfinity}$   
*uminus-MinusInfinity*:  $- \text{MinusInfinity} = \text{Infinity}$   
*uminus-Nullity*:  $- \text{Nullity} = \text{Nullity}$

A6

**defs (overloaded)**

*trans-number-subtract-def*:  
 $!! \ (y::'a::\{\text{plus}, \text{minus}\} \text{ trans-number}).$   
 $x - y == x + (- \ y)$

**lemmas** *trans-number-defs* =

*trans-number-zero-def* *trans-number-infinity-def*  
*trans-number-nullity-def*  
*trans-number-subtract-def*

**lemma** *trans-number-minus-infinity-sym-def*:  $\text{MinusInfinity} == -\infty$   
**by** (*simp add: trans-number-defs*)

**lemmas** *trans-number-sym-defs* =

*trans-number-zero-def* [*symmetric*]  
*trans-number-infinity-def* [*symmetric*]  
*trans-number-minus-infinity-sym-def*  
*trans-number-nullity-def* [*symmetric*]  
*trans-number-subtract-def* [*symmetric*]

**lemma** *primitive-zero*[simp]: *primitive 0*  
**by** (*unfold trans-number-zero-def, simp*)

**lemma** *not-primitive-simps*[simp]:  
 $\neg \text{primitive } (\infty) \wedge \neg \text{primitive } (-\infty) \wedge \neg \text{primitive } \Phi$   
**by** (*unfold trans-number-defs, simp*)

**lemma** *primitive-iff*:  
 $\text{primitive } x = (x \neq \infty \wedge x \neq -\infty \wedge x \neq \Phi)$   
**by** (*case-tac x, simp-all add: trans-number-defs*)

#### 4.1 Distinctness of special values

**lemma** *P-neq-infinity*[simp]:  $P \ x \neq \infty$   
**by** (*unfold trans-number-defs, simp*)

**lemma** *P-neq-minus-infinity*[simp]:  $P \ x \neq -\infty$   
**by** (*unfold trans-number-defs, simp*)

**lemma** *P-neq-nullity*[simp]:  $P \ x \neq \Phi$   
**by** (*unfold trans-number-defs, simp*)

**lemma** *zero-neq-infinity*[simp]:  $(0::'a::\text{zero trans-number}) \neq \infty$   
**by** (*unfold trans-number-defs, simp*)

**lemma** *zero-neq-minus-infinity*[simp]:  
 $(0::'a::\{\text{zero, minus}\} \text{ trans-number}) \neq -\infty$   
**by** (*unfold trans-number-defs, simp*)

**lemma** *zero-neq-nullity*[simp]:  $(0::'a::\{\text{zero, minus}\} \text{ trans-number}) \neq \Phi$   
**by** (*unfold trans-number-defs, simp*)

**lemma** *infinity-neq-minus-infinity*[simp]:  $(\infty::'a::\text{minus trans-number}) \neq -\infty$   
**by** (*unfold trans-number-defs, simp*)

**lemma** *infinity-neq-nullity*[simp]:  $(\infty::'a::\text{minus trans-number}) \neq \Phi$   
**by** (*unfold trans-number-defs, simp*)

**lemma** *minus-infinity-neq-nullity*[simp]:  $(-\infty::'a::\text{minus trans-number}) \neq \Phi$   
**by** (*unfold trans-number-defs, simp*)

**declare** *P-neq-infinity* [*THEN not-sym, simp*]  
**declare** *P-neq-minus-infinity* [*THEN not-sym, simp*]  
**declare** *P-neq-nullity* [*THEN not-sym, simp*]  
**declare** *zero-neq-infinity* [*THEN not-sym, simp*]  
**declare** *zero-neq-minus-infinity* [*THEN not-sym, simp*]  
**declare** *zero-neq-nullity* [*THEN not-sym, simp*]  
**declare** *infinity-neq-minus-infinity* [*THEN not-sym, simp*]

**declare** *infinity-neq-nullity* [*THEN not-sym,simp*]  
**declare** *minus-infinity-neq-nullity* [*THEN not-sym,simp*]

**lemma** *P-add-P*[*simp*]:  $P\ x + P\ y = P\ ((x :: 'a :: \{plus, minus\}) + y)$   
**by** (*simp add: trans-number-subtract-def*)

**lemma** *P-add-non-primitive*[*simp*]:  $P\ x + \infty = \infty \wedge P\ x - \infty = -\infty \wedge P\ x + \Phi = \Phi$   
**by** (*simp add: trans-number-subtract-def*)

**lemma** *Infinity-add-left*[*simp*]:  
 $\infty + P\ x = (\infty :: 'a :: \{plus, minus\}\ trans-number) \wedge$   
 $\infty + \infty = (\infty :: 'a\ trans-number) \wedge$   
 $\infty - \infty = (\Phi :: 'a\ trans-number) \wedge$   
 $\infty + \Phi = (\Phi :: 'a\ trans-number)$   
**by** (*simp add: trans-number-defs*)

**lemma** *MinusInfinity-add-left*[*simp*]:  
 $-\infty + P\ x = (-\infty :: 'a :: \{plus, minus\}\ trans-number) \wedge$   
 $-\infty + \infty = (\Phi :: 'a\ trans-number) \wedge$   
 $-\infty - \infty = (-\infty :: 'a\ trans-number) \wedge$   
 $-\infty + \Phi = (\Phi :: 'a\ trans-number)$   
**by** (*simp add: trans-number-defs*)

**lemma** *uminus-simps*[*simp*]:  
 $- P\ x = (P\ (-\ x) :: 'a :: \{minus\}\ trans-number) \wedge$   
 $-( -\infty) = (\infty :: 'a\ trans-number) \wedge$   
 $-\Phi = (\Phi :: 'a\ trans-number)$   
**by** (*simp add: trans-number-defs*)

A4

**lemma** *nullity-add-left*[*simp*]:  
 $\Phi + x = (\Phi :: 'a :: \{plus, minus\}\ trans-number)$   
**by** (*simp add: trans-number-nullity-def*)

**lemma** *nullity-subtract-left*[*simp*]:  
 $\Phi - x = (\Phi :: 'a :: \{plus, minus\}\ trans-number)$   
**by** (*simp add: trans-number-defs*)

Axiom A5

**lemma** *addition-infinity-not-null*:  
 $\llbracket x \neq -\infty ; x \neq \Phi \rrbracket \implies (x :: 'a :: \{plus, minus\}\ trans-number) + \infty = \infty$   
**by** (*case-tac x, simp-all add: trans-number-defs*)

A1



**lemma** *add-assoc*:

$((x::'a::ab\text{-group-add trans-number}) + y) + z = x + (y + z)$

**apply** (*case-tac* *x*, *safe*)  
**apply** (*tactic* *ALLGOALS* (*case-tac* *y*), *safe*)  
**apply** (*tactic* *ALLGOALS* (*case-tac* *z*), *safe*)  
**by** (*simp-all* *add: trans-number-sym-defs*)

A2

**lemma** *add-commute*:

$(x::'a::ab\text{-group-add trans-number}) + y = y + x$

**apply** (*case-tac* *x*, *safe*)  
**apply** (*tactic* *ALLGOALS* (*case-tac* *y*))  
**by** (*simp-all* *add: trans-number-sym-defs*)

**lemma** *add-left-commute*:

$a + (b + c) = b + (a + (c::'a::ab\text{-group-add trans-number}))$   
**by** (*rule* *mk-left-commute* [*of op +, OF add-assoc add-commute*])

**theorems** *trans-add-ac = add-assoc add-commute add-left-commute*

**lemma** *nullity-add-right[simp]*:

$x + \Phi = (\Phi::'a::ab\text{-group-add trans-number})$   
**by** (*subst* *add-commute*, *simp* *add: trans-number-nullity-def*)

A3

**lemma** *add-identity[simp]*:

$0 + (x::'a::ab\text{-group-add trans-number}) = x$

**apply** (*unfold* *trans-number-zero-def*, *case-tac* *x*)  
**by** (*simp-all*, *simp-all* *add: trans-number-defs*)

A7

**lemma** *bijectivity-of-uminus[simp]*:

$-(-(x::'a::ab\text{-group-add trans-number})) = x$

**by** (*case-tac* *x*, *simp-all* *add: trans-number-sym-defs*)

A8

**lemma** *additive-inverse[simp]*:

$!! (x::'a::ab\text{-group-add trans-number}).$   
 $\text{primitive } x \implies x - x = 0$

**apply** (*case-tac* *x*)  
**by** (*simp-all* *add: trans-number-subtract-def trans-number-zero-def*)

A9

```

lemma uminus-nullity:
  - ( $\Phi :: ('a::\text{minus}) \text{ trans-number}$ ) =  $\Phi$ 

by simp

A10

lemma subtraction-infinity-not-null:
  !! ( $x::'a::\{\text{plus},\text{minus}\} \text{ trans-number}$ ).
     $\llbracket x \neq \infty; x \neq \Phi \rrbracket \implies x - \infty = -\infty$ 

by (case-tac x, simp-all add: trans-number-defs)

instance trans-number :: (ab-group-add) trans-add
apply (intro-classes, simp-all add:)
apply (rule add-assoc [THEN sym])
apply (rule add-commute)
apply (subst add-commute, erule addition-infinity-not-null, assumption)
apply (simp add: trans-number-subtract-def)
apply (rule additive-inverse, simp add: primitive-iff)
by (erule subtraction-infinity-not-null, assumption)

```

## 5 Transnumber ordering

```

instance trans-number :: (ord) ord ..
primrec
  P-less:  $P\ x < y = (\text{primitive } y \wedge (x::'a::\{\text{minus},\text{ord}\}) < \text{inv } P\ y \mid y = \infty)$ 
  Infinity-less:  $(\text{Infinity} < (y::'a::\{\text{minus},\text{ord}\} \text{ trans-number})) = \text{False}$ 
  MinusInfinity-less:
     $(\text{MinusInfinity} < y) = (y \neq -\infty \wedge y \neq (\Phi::'a::\{\text{minus},\text{ord}\} \text{ trans-number}))$ 
  Nullity-less:  $(\text{Nullity} < (y::'a::\{\text{minus},\text{ord}\} \text{ trans-number})) = \text{False}$ 

defs (overloaded)
  trans-number-le-def:
     $(x::'a::\{\text{minus},\text{ord}\} \text{ trans-number}) \leq y == (x < y \mid x = y)$ 

lemma P-less-P[simp]:  $(P\ x < P\ y) = (x < (y::'a::\{\text{minus},\text{order}\}))$ 
by (simp add: trans-number-defs)

lemma P-less-non-primitive[simp]:  $(P\ x < \infty) \wedge \neg (P\ x < -\infty) \wedge \neg (P\ x < \Phi)$ 
by (simp add: trans-number-defs)

lemma P-le[simp]:
   $(P\ x \leq P\ y = (x \leq (y::'a::\{\text{order},\text{minus}\}))) \wedge$ 
   $(P\ x \leq \infty) \wedge$ 
   $\neg (P\ x \leq -\infty) \wedge$ 
   $\neg (P\ x \leq \Phi)$ 
by (simp add: trans-number-defs trans-number-le-def order-le-less)

lemma not-infinity-less[simp]:

```

$\neg ((\infty::'a::\{\text{minus}, \text{ord}\} \text{ trans-number}) < x)$   
**by** (*unfold trans-number-defs, simp*)

**lemma** *infinity-le [simp]*:  
 $((\infty::'a::\{\text{minus}, \text{ord}\} \text{ trans-number}) \leq x) = (x = \infty)$   
**by** (*auto simp: trans-number-le-def*)

**lemma** *le-infinity [simp]*:  $x \leq (\infty::'a::\{\text{minus}, \text{ord}\} \text{ trans-number}) = (x \neq \Phi)$   
**by** (*case-tac x, auto simp: trans-number-le-def trans-number-defs*)

**lemma** *minus-infinity-less [simp]*:  
 $(-(\infty::'a::\{\text{minus}, \text{ord}\} \text{ trans-number}) < x) = (x \neq -\infty \wedge x \neq \Phi)$   
**by** (*simp add: trans-number-defs*)

**lemma** *minus-infinity-le [simp]*:  
 $(-(\infty::'a::\{\text{minus}, \text{ord}\} \text{ trans-number}) \leq x) = (x \neq \Phi)$   
**by** (*auto simp: trans-number-le-def*)

**lemma** *not-nullity-less [simp]*:  $\neg ((\Phi::'a::\{\text{minus}, \text{ord}\} \text{ trans-number}) < x)$   
**by** (*unfold trans-number-nullity-def, simp*)

**lemma** *nullity-le [simp]*:  $((\Phi::'a::\{\text{minus}, \text{ord}\} \text{ trans-number}) \leq x) = (x = \Phi)$   
**by** (*auto simp: trans-number-le-def*)

**lemma** *not-less-nullity [simp]*:  $\neg (x < (\Phi::'a::\{\text{minus}, \text{ord}\} \text{ trans-number}))$   
**by** (*case-tac x, simp-all*)

**lemma** *le-nullity [simp]*:  
 $(x \leq (\Phi::'a::\{\text{minus}, \text{ord}\} \text{ trans-number})) = (x = \Phi)$   
**by** (*case-tac x, simp-all add: trans-number-sym-defs*)

**lemma** *zero-less-P [simp]*:  
 $((0::'a::\{\text{minus}, \text{ord}, \text{zero}\} \text{ trans-number}) < P x) = (0 < x)$   
**by** (*simp add: trans-number-zero-def*)

A25

**lemma** *zero-less-infinity [simp]*:  $((0::'a::\{\text{minus}, \text{ord}, \text{zero}\} \text{ trans-number}) < \infty)$   
**by** (*simp add: trans-number-zero-def*)

**lemma** *zero-not-less-minus-infinity [simp]*:  
 $\neg ((0::'a::\{\text{minus}, \text{ord}, \text{zero}\} \text{ trans-number}) < -\infty)$   
**by** (*simp add: trans-number-zero-def*)

**lemma** *irreflexive-less [simp]*:  $\neg ((x::'a::\{\text{minus}, \text{ord}, \text{zero}\} \text{ trans-number}) < x)$   
**by** (*case-tac x, auto simp: trans-number-sym-defs*)

**declare** *P-less [simp del]*

A26

**lemma** *trans-number-ordering*:  
 $(x - y > 0) = (x > (y :: 'a :: \text{ordered-ab-group trans-number}))$   
**apply** (*case-tac*  $x$ )  
**apply** (*tactic ALLGOALS* (*case-tac*  $y$ ))  
**apply** (*simp-all* *add: trans-number-subtract-def*)  
**apply** (*unfold* *trans-number-sym-defs*)  
**by** (*simp-all* *add: compare-rls*)

A27 holds trivially since *gt* is simply a syntactic abbreviation

**lemma** *less-than-gt-than-eq*:  $(x < y) = (y > x)$   
**by** (*rule refl*)

A28

**lemma** *quadrachotomy*:  
 $!!x :: 'a :: \{ab\text{-group-add}, \text{linorder}\} \text{ trans-number.}$   
 $x \text{ or } [x < 0, x = 0, 0 < x, x = \Phi]$   
**apply** (*case-tac*  $x$ )  
**apply** (*simp-all* *add: trans-number-sym-defs*)  
**apply** (*rule-tac*  $x=a$  **and**  $y=0$  **in** *linorder-cases*)  
**by** (*auto simp: trans-number-zero-def*)

A29

**lemma** *pos-closure-add*:  
 $!! x :: 'a :: \{\text{ordered-ab-group}\} \text{ trans-number.}$   
 $\llbracket 0 < x ; 0 < y \rrbracket \implies 0 < x + y$   
**apply** (*case-tac*  $x$ )  
**apply** (*tactic ALLGOALS* (*case-tac*  $y$ ))  
**apply** (*simp-all* *add: trans-number-sym-defs*)  
**by** (*erule* *add-pos-pos*, *assumption*)

**lemma** *trans-number-order-refl*:  
 $(x :: 'a :: \{\text{minus}, \text{order}\} \text{ trans-number}) \leq x$   
**apply** (*case-tac*  $x$ )  
**by** (*simp-all* *add: trans-number-sym-defs*)

**lemma** *trans-number-order-trans*:  
 $\llbracket (x :: 'a :: \{\text{minus}, \text{order}\} \text{ trans-number}) \leq y ; y \leq z \rrbracket \implies x \leq z$   
**apply** (*case-tac*  $x$ )  
**apply** (*tactic ALLGOALS* (*case-tac*  $y$ ))  
**apply** (*tactic ALLGOALS* (*case-tac*  $z$ ))  
**by** (*simp-all* *add: trans-number-sym-defs*)

**lemma** *trans-number-order-antisym*:  
 $\llbracket (x :: 'a :: \{\text{minus}, \text{order}\} \text{ trans-number}) \leq y ; y \leq x \rrbracket \implies x = y$

```

apply (case-tac x)
apply (tactic ALLGOALS (case-tac y))
by (simp-all add: trans-number-sym-defs)

lemma trans-number-less-le:
  ((x::'a::{minus,order} trans-number) < y) = (x ≤ y ∧ x ≠ y)

apply (unfold trans-number-le-def, auto)
by (case-tac y, simp-all add: trans-number-sym-defs)

axclass minus-order ⊆ order, minus

instance pordered-ab-group-add ⊆ minus-order ..

instance trans-number :: (minus-order)order
apply (intro-classes)
apply (rule trans-number-order-refl)
apply (erule trans-number-order-trans, assumption)
apply (erule trans-number-order-antisym, assumption)
by (rule trans-number-less-le)

lemma ext-number-linear:
  [| (x::'a::{minus,linorder} trans-number) ≠ Φ; y ≠ Φ |] ⇒ x ≤ y | y ≤ x
apply (case-tac x)
apply (tactic ALLGOALS (case-tac y))
by (simp-all add: trans-number-sym-defs linorder-linear)

lemma not-elem-conv: xs ⊆ {x. x ≠ a} = (a ∉ xs)
by (unfold subset-def, auto)

```

## 5.1 Lattice-completeness of trans\_numbers

NOTE: cannot express l.-c. of transnumbers as instance rule

```

lemma ext-number-complete-lattice:
  lattice-complete {x :: real trans-number. x ≠ Φ}

apply (unfold lattice-complete-def)
apply (clarsimp simp: not-elem-conv)
apply (case-tac ∃ a. ∀ x ∈ ys. x < P a)
apply (case-tac ∃ b. P b ∈ ys, safe)
apply (cut-tac S = inv P ' (ys - {-∞}) in reals-complete)
apply (fastsimp)
apply (rule-tac x=a in exI)
apply (unfold isUb-def settle-def isLub-def leastP-def setge-def, safe)
apply (case-tac x, unfold trans-number-sym-defs)
  apply (fastsimp, fastsimp, fastsimp, fastsimp, fast)
apply (clarsimp simp: isUb-def settle-def)
apply (rule-tac x=P t in exI, safe, simp-all)
apply (case-tac y, unfold trans-number-sym-defs, simp)

```

```

apply (erule-tac  $x = P$  aa in ballE) back
apply (simp+, force, fastsimp, fast)
apply (case-tac v, unfold trans-number-sym-defs)
apply clarsimp
apply (erule-tac  $x = aa$  in allE, clarsimp)
apply (case-tac y, unfold trans-number-sym-defs)
apply (fastsimp, fastsimp, assumption, fastsimp, simp)
apply (fastsimp, fastsimp)
apply (erule-tac  $x = -\infty$  in exI, clarsimp)
apply (case-tac y, unfold trans-number-sym-defs)
apply (fastsimp, fastsimp, simp, fastsimp)
apply (erule-tac  $x = \infty$  in exI, auto)
apply (case-tac v, unfold trans-number-sym-defs)
apply (auto)
apply (erule-tac  $x = a + 1$  in allE, clarify)
apply (erule-tac  $x = x$  in ballE)
apply (auto simp: trans-number-less-le)
apply (erule-tac notE, erule-tac trans-number-order-trans, simp)
apply (erule-tac  $x = 1$  in allE, auto)
apply (erule-tac  $x = x$  in ballE)
by (erule-tac notE, erule-tac trans-number-order-trans, fastsimp+)

```

## 6 A model for axiomatic class trans\_mult

```

instance trans-number :: (one) one ..
instance trans-number :: (inverse) inverse ..
instance trans-number :: (times) times ..

defs (overloaded)
  trans-number-one-def:  $1 == P\ 1$ 

lemmas trans-number-defs =
  trans-number-zero-def trans-number-one-def
  trans-number-infinity-def trans-number-nullity-def
  trans-number-subtract-def

lemmas trans-number-sym-defs =
  trans-number-zero-def [symmetric]
  trans-number-one-def [symmetric]
  trans-number-infinity-def [symmetric]
  trans-number-minus-infinity-sym-def
  trans-number-nullity-def [symmetric]
  trans-number-subtract-def [symmetric]

lemma primitive-one[simp]: primitive 1
  by (unfold trans-number-one-def, simp)

```

Warning: simpsets contain different mult laws with special cases for RHS

**primrec**

*trans-mult-P*:

$P (x :: 'a :: \{\text{zero}, \text{times}, \text{minus}, \text{ord}\}) * y =$   
     ( if primitive  $y$  then  $P (x * \text{inv } P y)$  else  
       if  $(y = \infty \wedge x > 0) \mid (y = -\infty \wedge x < 0)$  then  $\infty$  else  
       if  $(y = \infty \wedge x < 0) \mid (y = -\infty \wedge x > 0)$  then  $-\infty$   
       else  $\Phi$ )

*trans-mult-Infinity*:

$\text{Infinity} * (y :: 'a :: \{\text{zero}, \text{times}, \text{minus}, \text{ord}\} \text{ trans-number}) =$   
     ( if (primitive  $y \wedge y > 0$ )  $\mid y = \infty$  then  $\infty$  else  
       if (primitive  $y \wedge y < 0$ )  $\mid y = -\infty$  then  $-\infty$   
       else  $\Phi$ )

*trans-mult-MinusInfinity*:

$\text{MinusInfinity} * (y :: 'a :: \{\text{zero}, \text{times}, \text{minus}, \text{ord}\} \text{ trans-number}) =$   
     ( if (primitive  $y \wedge y < 0$ )  $\mid y = -\infty$  then  $\infty$  else  
       if (primitive  $y \wedge y > 0$ )  $\mid y = \infty$  then  $-\infty$   
       else  $\Phi$ )

*trans-mult-Nullity*:  $\text{Nullity} * (y :: 'a :: \{\text{zero}, \text{times}, \text{minus}, \text{ord}\} \text{ trans-number}) = \Phi$

**lemma** *P-mult-P[simp]*:

$P x * P y = P ((x :: 'a :: \{\text{zero}, \text{times}, \text{minus}, \text{linorder}\}) * y)$

**by** *auto*

A15 is first conjunct of `mult_nullity`, see also `trans_mult_nullity`

**lemma** *mult-nullity[simp]*:

$(\Phi :: 'a :: \text{ordered-idom trans-number}) * x = \Phi \wedge x * \Phi = \Phi$

**apply** (*rule conjI*)

**apply** (*simp add: trans-number-nullity-def*)

**by** (*case-tac x, auto*)

A16 is fourth conjunct of `mult_zero`

**lemma** *mult-zero[simp]*:

$0 * P (x :: 'a :: \text{ordered-idom}) = 0 \wedge$   
 $P x * 0 = 0 \wedge$   
 $(0 :: 'a \text{ trans-number}) * \infty = \Phi \wedge$   
 $\infty * (0 :: 'a \text{ trans-number}) = \Phi \wedge$   
 $(0 :: 'a \text{ trans-number}) * -\infty = \Phi \wedge$   
 $-\infty * (0 :: 'a \text{ trans-number}) = \Phi \wedge$   
 $\infty * \Phi = (\Phi :: 'a \text{ trans-number})$

**by** (*simp add: trans-number-defs*)

**lemma** *mult-infinity[simp]*:

$\infty * (\infty :: 'a :: \text{ordered-idom trans-number}) = \infty \wedge$   
 $(\infty :: 'a \text{ trans-number}) * -\infty = -\infty \wedge$

$-\infty * (\infty :: 'a \text{ trans-number}) = -\infty \wedge$   
 $-\infty * -\infty = (\infty :: 'a \text{ trans-number})$

**by** (*simp add: trans-number-defs*)

**lemma** *P-mult-infinity-less-zero*:

!!  $x :: ('a :: \text{ordered-idom})$ .

$x < 0 \implies P\ x * \infty = -\infty \wedge \infty * P\ x = -\infty$

**by** (*simp add: trans-number-defs*)

**lemma** *P-mult-infinity-gt-zero*:

!!  $x :: ('a :: \text{ordered-idom})$ .

$0 < x \implies P\ x * \infty = \infty \wedge \infty * P\ x = \infty$

**by** (*simp add: trans-number-defs*)

**lemma** *P-mult-MinusInfinity-less-zero*:

!!  $x :: ('a :: \text{ordered-idom})$ .

$x < 0 \implies P\ x * -\infty = \infty \wedge -\infty * P\ x = \infty$

**by** (*simp add: trans-number-defs*)

**lemma** *P-mult-MinusInfinity-gt-zero*:

!!  $x :: ('a :: \text{ordered-idom})$ .

$0 < x \implies P\ x * -\infty = -\infty \wedge -\infty * P\ x = -\infty$

**by** (*simp add: trans-number-defs*)

**lemmas** *P-mult-infinities =*

*P-mult-infinity-less-zero P-mult-infinity-gt-zero*

*P-mult-MinusInfinity-less-zero P-mult-MinusInfinity-gt-zero*

**declare** *trans-mult-P* [*simp del*]

*trans-mult-Infinity* [*simp del*]

*trans-mult-MinusInfinity* [*simp del*]

*trans-mult-Nullity* [*simp del*]

A13

**lemma** *mult-commute*:

$(x :: 'a :: \text{ordered-idom trans-number}) * y = y * x$

**apply** (*case-tac x, safe*)

**apply** (*tactic ALLGOALS (case-tac y), safe*)

**apply** (*simp-all add: trans-number-sym-defs mult-ac*)

**by** (*((rule-tac x=a and y=0 in linorder-cases,*

*(simp add: trans-number-zero-def [symmetric] P-mult-infinities)+)+)*

A12



**lemma** *mult-assoc*:

$((x::'a::\text{ordered-idom } \text{trans-number}) * y) * z = x * (y * z)$

**apply** (*case-tac* *x*, *safe*)  
**apply** (*tactic* *ALLGOALS* (*case-tac* *y*), *safe*)  
**apply** (*tactic* *ALLGOALS* (*case-tac* *z*), *safe*)  
**apply** (*simp-all* *add: trans-number-sym-defs mult-ac*)  
**apply** (*tactic* *ALLGOALS* (*case-tac* *a = 0*))  
**apply** (*tactic* *ALLGOALS* (*case-tac* *a < 0*))  
**apply** (*simp-all* *add: trans-number-sym-defs*  
*P-mult-infinities mult-ac linorder-not-less order-le-less*)  
**apply** (*tactic* *ALLGOALS*(*case-tac* *aa = 0*))  
**apply** (*tactic* *ALLGOALS*(*case-tac* *aa < 0*))  
**by** (*simp-all* *add: trans-number-sym-defs*  
*P-mult-infinities mult-ac linorder-not-less order-le-less*  
*zero-less-mult-iff mult-less-0-iff*)

**lemma** *mult-left-commute*:

$x * (y * z) = y * (x * (z::'a:: \text{ordered-idom } \text{trans-number}))$   
**by** (*rule* *mk-left-commute* [*of op \**, *OF mult-assoc mult-commute*])

**lemmas** *trans-mult-ac = mult-assoc mult-commute mult-left-commute*

A14

**lemma** *mult-one-left[simp]*:  $1 * x = (x::'a:: \text{ordered-idom } \text{trans-number})$

**apply** (*case-tac* *x*)  
**apply** (*simp-all* *add: trans-number-one-def*)  
**apply** (*simp* *add: P-mult-infinities trans-number-infinity-def[symmetric]*)  
**apply** (*simp* *add: P-mult-infinities*  
*uminus-Infinity [THEN sym] trans-number-infinity-def[symmetric]*)  
**by** (*simp* *add: trans-number-nullity-def [symmetric]*)

**lemma** *mult-one-right[simp]*:  $x * 1 = (x::'a:: \text{ordered-idom } \text{trans-number})$

**by** (*subst* *mult-commute*, *rule* *mult-one-left*)

**lemma** *not-primitive-mult-infinity[simp]*:

$\neg (\text{primitive } (P (x::'a::\text{ordered-idom}) * \infty))$   
**apply** (*case-tac* *x < 0*)  
**apply** (*simp* *add: P-mult-infinities*)  
**by** (*auto* *simp: linorder-not-less order-le-less*  
*P-mult-infinities trans-number-sym-defs*)

**lemma** *not-primitive-mult-MinusInfinity[simp]*:

$\neg (\text{primitive } (P (x::'a::\text{ordered-idom}) * -\infty))$   
**apply** (*case-tac* *x < 0*)  
**apply** (*simp* *add: P-mult-infinities*)  
**by** (*auto* *simp: linorder-not-less order-le-less*  
*P-mult-infinities trans-number-sym-defs*)

## 6.1 Inverse and division

**primrec**

$\text{inverse } (P \ (x::'a::\{\text{inverse}, \text{zero}\})) = (\text{if } x = 0 \text{ then } \infty \text{ else } P \ (\text{inverse } x))$   
 $\text{inverse } (\text{Infinity}::('a::\{\text{inverse}, \text{zero}\}) \text{ trans-number}) = 0$   
 $\text{inverse } (\text{MinusInfinity}::('a::\{\text{inverse}, \text{zero}\}) \text{ trans-number}) = 0$   
 $\text{inverse } (\text{Nullity}::('a::\{\text{inverse}, \text{zero}\}) \text{ trans-number}) = \Phi$

A17

**defs (overloaded)**

$\text{trans-number-divison-def:}$   
 $x / (y::'a::\{\text{times}, \text{inverse}\} \text{ trans-number}) == x * \text{inverse } y$

**lemma inverse-Infinity[simp]:**

$\text{inverse } (\infty :: ('a::\{\text{inverse}, \text{zero}\}) \text{ trans-number}) = 0$

**by** (simp add: trans-number-defs)

**lemma inverse-MinusInfinity[simp]:**

$\text{inverse } (-\infty :: ('a::\{\text{inverse}, \text{zero}, \text{minus}\}) \text{ trans-number}) = 0$

**by** (simp add: trans-number-defs)

**lemma inverse-nullity[simp]:**

$\text{inverse } (\Phi :: ('a::\{\text{inverse}, \text{zero}\}) \text{ trans-number}) = \Phi$

**by** (simp add: trans-number-defs)

A18

**lemma multiplicative-inverse:**

$\llbracket \text{primitive } (x::'a::\text{ordered-field trans-number}); x \neq 0 \rrbracket \implies x / x = 1$

**apply** (case-tac x, simp-all)

**by** (auto simp: trans-number-divison-def trans-number-sym-defs)

A19

**lemma bij-inverse:**

$(x::'a::\text{ordered-field trans-number}) \neq -\infty \implies \text{inverse } (\text{inverse } x) = x$

**apply** (case-tac x)

**apply** (simp-all add: trans-number-defs nonzero-inverse-inverse-eq)

**by** (fast intro: inverse-zero-imp-zero)

A20

**lemma inverse-zero[simp]:**

$\text{inverse } (0::'a::\text{ordered-field trans-number}) = \infty$

**by** (simp add: trans-number-defs)

A21

**lemma** *inverse-MinusInfinity*[simp]:  
 $inverse \ (-\infty :: 'a::ordered-field \ trans-number) = 0$   
**by** (simp add: trans-number-defs)

A22

**lemma** *inverse-nullity*[simp]:  
 $inverse \ (\Phi :: 'a::ordered-field \ trans-number) = \Phi$   
**by** (simp add: trans-number-defs)

A23

**lemma** *positive-inf-mult*:  
 $(\infty * x = \infty) = (0 < (x :: 'a::ordered-field \ trans-number))$   
**apply** (case-tac x)  
**apply** (simp-all add: trans-number-sym-defs)  
**apply** (rule-tac  $x=a$  **and**  $y=0$  **in** linorder-cases)  
**apply** (simp-all add: P-mult-infinities)  
**by** (simp add: trans-number-sym-defs)

A24

**lemma** *negative-inf-mult*:  
 $(\infty * x = -\infty) = (x < (0 :: 'a::ordered-field \ trans-number))$   
**apply** (case-tac x)  
**apply** (simp-all add: trans-number-sym-defs)  
**apply** (simp add: trans-number-zero-def)  
**apply** (rule-tac  $x=a$  **and**  $y=0$  **in** linorder-cases)  
**apply** (simp-all add: P-mult-infinities)  
**by** (simp add: trans-number-sym-defs)

**instance** *trans-number* :: (ordered-idom) sgn ..

**defs** (overloaded)  
*trans-number-sgn-def*:  
 $sgn \ (a :: 'a::ordered-idom \ trans-number)$   
 $== (if \ 0 < a \ then \ 1 \ else$   
 $\quad if \ 0 = a \ then \ 0 \ else$   
 $\quad if \ a < 0 \ then \ - \ 1 \ else$   
 $\quad (* \ a = \Phi *) \quad \Phi)$

**instance** *trans-number* :: (ordered-idom) trans-sgn  
**by** (intro-classes, unfold trans-number-sgn-def, rule refl)

**lemma** *P-mult-infinity-neq-infinity-iff*:  
 $(P \ a * \infty \neq \infty) = (a \leq (0 :: 'a::ordered-idom))$

**apply** (case-tac  $a < 0$ , simp add:  $P\text{-mult-infinity-less-zero}$ )  
**by** (auto simp:  $P\text{-mult-infinity-gt-zero}$   
linorder-not-less order-le-less trans-number-sym-defs)

**lemma**  $P\text{-mult-infinity-neq-MinusInfinity-iff}$ :  
 $(P\ a * \infty \neq -\infty) = (0 \leq (a::'a::\text{ordered-idom}))$   
**apply** (case-tac  $a < 0$ , simp add:  $P\text{-mult-infinity-less-zero}$ )  
**by** (auto simp:  $P\text{-mult-infinity-gt-zero}$   
linorder-not-less order-le-less trans-number-sym-defs)

**lemma**  $P\text{-mult-MinusInfinity-neq-MinusInfinity-iff}$ :  
 $(P\ a * -\infty \neq -\infty) = (a \leq (0::'a::\text{ordered-idom}))$   
**apply** (case-tac  $0 < a$ , simp add:  $P\text{-mult-MinusInfinity-gt-zero}$ )  
**by** (auto simp:  $P\text{-mult-MinusInfinity-less-zero}$   
linorder-not-less order-le-less trans-number-sym-defs)

**lemma**  $P\text{-mult-MinusInfinity-neq-infinity-iff}$ :  
 $(P\ a * -\infty \neq \infty) = (0 \leq (a::'a::\text{ordered-idom}))$   
**apply** (case-tac  $0 < a$ , simp add:  $P\text{-mult-MinusInfinity-gt-zero}$ )  
**by** (auto simp:  $P\text{-mult-MinusInfinity-less-zero}$   
linorder-not-less order-le-less trans-number-sym-defs)

**lemma**  $\text{sgn-}P$ :  
 $\text{sgn}\ (P\ (x::'a::\text{ordered-idom})) = (\text{if } x < 0 \text{ then } -1 \text{ else if } x = 0 \text{ then } 0 \text{ else } 1)$   
**apply** (unfold trans-number-sgn-def, simp add: trans-number-sym-defs)  
**by** (safe, simp-all add: trans-number-zero-def)

**lemma**  $\text{uminus-eq-iff}$ :  
 $(-x = (x::'a::\text{ordered-idom})) = (x = 0)$   
**by** auto

**lemma**  $\text{sgn-}P\text{-eq-iff}$ :  
 $!! (x::'a::\text{ordered-idom})\ (y::'a).$   
 $(\text{sgn}\ (P\ x) = \text{sgn}\ (P\ y))$   
 $= (((x < 0) = (y < 0)) \wedge ((x = 0) = (y = 0)) \wedge$   
 $((0 < x) = (0 < (y::'a::\text{ordered-idom}))))$   
**apply** (simp add:  $\text{sgn-}P$ , safe, simp-all)  
**apply** (simp add: trans-number-defs)  
**apply** (simp add: trans-number-defs)  
**apply** (simp add: trans-number-defs)  
**apply** (unfold trans-number-defs)  
**apply** (simp only:  $\text{uminus-}P$  trans-number.simps  $\text{uminus-eq-iff}$ )  
**apply** (simp)  
**by** (simp only:  $\text{uminus-}P$  trans-number.simps  $\text{uminus-eq-iff}$ )

**lemma**  $\text{sgn-zero[simp]}$ :  $\text{sgn}\ (0::('a::\text{ordered-idom}\ \text{trans-number})) = 0$   
**by** (simp add: trans-number-sgn-def)

**lemma** *sgn-infinity*[simp]:  $\text{sgn } (\infty :: ('a::\text{ordered-idom trans-number})) = 1$   
**by** (*simp add: trans-number-sgn-def*)

**lemma** *sgn-minus-infinity*[simp]:  $\text{sgn } (-\infty :: ('a::\text{ordered-idom trans-number})) = -1$   
**by** (*simp add: trans-number-sgn-def*)

**lemma** *sgn-zero-iff*[simp]:  
 $(\text{sgn } (x :: ('a::\text{ordered-idom trans-number})) = 0) = (x = 0)$   
**by** (*auto simp: trans-number-sgn-def, simp-all add: trans-number-defs*)

**lemma** *sgn-P-one-iff*[simp]:  
 $(\text{sgn } (P (x :: ('a::\text{ordered-idom}))) = 1) = (0 < x)$   
**apply** (*simp add: trans-number-sgn-def, safe, simp*)  
**apply** (*unfold trans-number-defs*)  
**apply** (*simp only: uminus-P trans-number.simps uminus-eq-iff*)  
**by** *simp-all*

**lemma** *sgn-P-minus-one-iff*[simp]:  
 $(\text{sgn } (P (x :: ('a::\text{ordered-idom}))) = -1) = (x < 0)$   
**apply** (*simp add: trans-number-sgn-def, safe, simp*)  
**apply** (*unfold trans-number-defs*)  
**apply** (*fast, simp, simp*)  
**apply** (*simp only: uminus-P trans-number.simps uminus-eq-iff*)  
**apply** (*blast intro: order-less-asm, simp*)  
**apply** (*simp only: uminus-P trans-number.simps uminus-eq-iff*)  
**apply** (*blast intro: order-less-asm, simp, simp*)  
**apply** (*simp only: uminus-P trans-number.simps uminus-eq-iff*)  
**apply** (*blast intro: order-less-asm*)  
**by** (*simp, simp*)

**lemma** *P-eq-zero*:  $(P x = 0) = (x = 0)$   
**by** (*unfold trans-number-zero-def, auto*)

A29

**lemma** *distributivity*:  
 $!! a :: ('a::\text{ordered-field trans-number}).$   
 $\neg ((a = \infty \vee a = -\infty) \wedge \text{sgn } b \neq \text{sgn } c \wedge (b + c \notin \{0, \Phi\}))$   
 $\implies a * (b + c) = (a * b) + (a * c)$   
**apply** (*case-tac a, tactic ALLGOALS Clarsimp-tac*)  
**apply** (*tactic ALLGOALS (case-tac b) THEN ALLGOALS Clarsimp-tac*)  
**apply** (*tactic ALLGOALS (case-tac c) THEN ALLGOALS Clarsimp-tac*)  
**apply** (*simp-all add: trans-number-sym-defs*)  
**apply** (*rule right-distrib, safe*)  
**apply** (*simp add: P-mult-infinities trans-number-sym-defs*  
 $P\text{-mult-infinity-neq-infinity-iff } P\text{-mult-infinity-neq-MinusInfinity-iff}$ )  
**apply** (*simp add: P-mult-infinities trans-number-sym-defs*  
 $P\text{-mult-MinusInfinity-neq-infinity-iff } P\text{-mult-MinusInfinity-neq-MinusInfinity-iff}$ )

```

apply (tactic ALLGOALS (case-tac aa < 0))
apply (simp-all add: linorder-not-less order-le-less, safe)
apply (simp-all add: trans-number-sym-defs
      P-mult-infinities sgn-P-eq-iff P-eq-zero)
apply auto
apply (subst P-mult-infinity-less-zero [THEN conjunct2], simp, rule refl)
apply (subst P-mult-infinity-gt-zero [THEN conjunct2], simp, rule refl)
apply (drule sym) prefer 4 apply (drule sym)
by (simp-all add: P-mult-infinities trans-number-sym-defs)

instance trans-number :: (ordered-field) trans-mult
apply (intro-classes)
prefer 17
apply (rule quadrachotomy)
apply (simp-all add: multiplicative-inverse primitive-iff bij-inverse
      positive-inf-mult negative-inf-mult trans-number-ordering)
apply (rule mult-assoc [THEN sym])
apply (rule mult-commute)
apply (simp add: trans-number-divison-def)
apply (rule distributivity, simp)
by (auto simp: order-le-less)

end

```